SqueakでPromises/A+をやってみる

# はじめに
今年もやってきましたSmalltalk Advent Calendar。1年ぶりの投稿になります。
年末に引っ越しが控えているため、いつものように自作のライブラリ紹介という訳にはいきません。シンプルな小ネタでいこうと思います。

# SqueakがPromises/A+に対応?
普段は仕事で[Pharo](https://pharo.org/)ばかり使っているのですが、自称Smalltalkエバンジェリストなので、それ以外のSmalltalk処理系のことも常に気になっています。

10/14日に古豪のOSSのSmalltalk処理系、[Squeak](https://squeak.org)が久しぶりに[バージョンアップした](http://lists.squeakfoundation.org/pipermail/squeak-dev/2018-October/200231.html)とのことで、[リリースノート](https://raw.githubusercontent.com/squeak-smalltalk/squeak-app/squeak-trunk/release-notes/5.2)を見ていたのですが、奇妙な記述を見つけました。

> "Changed the Promise semantic to be closer to Javascript:"

PromiseをJavaScriptのセマンティクスに近づけた? ちょっと何を言っているかよくわからない...

そこで面白そうなので実際に動かしてみることにしました。

対話環境がSmalltalkの醍醐味です。以後は皆さんもご一緒にどうぞ。

# Squeak 5.2 のインストール
https://squeak.org/downloads/
から好きなプラットフォーム用のインストーラを入手します。面倒な人には[All-in-One(64-bit)](http://files.squeak.org/5.2/Squeak5.2-18225-64bit/Squeak5.2-18225-64bit-All-in-One.zip)がお勧めです。

zipファイルを展開すると起動用スクリプトがあります。Windowsであれば`squeak.bat`、それ以外は`squeak.sh`を使います。

モダンなのかレトロなのかよくわからない画面が立ち上がります。
![welcome.png](https://qiita-image-store.s3.amazonaws.com/0/61787/95220e6b-4020-6210-6b87-32d618802881.png)

"Configure"を選ぶとテーマの設定など、いろいろできるようですが、年末で忙しいのでここは華麗に"Skip"しましょう。

デフォルトだとフォントが若干小さすぎるため、気になる人はデスクトップメニュー(左クリック)から"appearance..."->"system fonts..."->"demo/hi-dpi mode"で大きくしておくと良いでしょう。

![font-menu.png](https://qiita-image-store.s3.amazonaws.com/0/61787/52690270-f51e-3aa1-5526-68fe9ec403f1.png)

# Promiseを見てみる

真ん中にどーんと表示されている"Welcome to Squeak"のウィンドウですが、実はここからリリースノートなどが読めるようになっています。"Release Notes"->"5.2"を選びスクロールしていくと例のフレーズ"Changed the Promise semantic to be closer to Javascript:"があり、Promiseクラスへのリンクが張られています。

![promise-link.png](https://qiita-image-store.s3.amazonaws.com/0/61787/108e7301-9ea0-69eb-b9dd-56c0fe65df35.png)

クリックするとPromiseのブラウザが開きます。

普段Smalltalkerはソース重視なのでコメントはほとんど読みませんが、今回は"instance"、"class"ボタンの隣にある小さな"?"ボタンを押してみましょう。

クラスのコメントが表示されます。

![comment2.png](https://qiita-image-store.s3.amazonaws.com/0/61787/54a4ec09-a539-ca29-c4c3-5195b8c4a48c.png)

```smalltalk

I also implement the Promises/A+ Javascript specification.
```

うーん本当? 期待して良いのでしょうか。

今度は右クリックメニューで"class refs"を選んで、参照元を見てみましょう。Alt+shift+nがショートカットです。(Macの場合はCommand+shift+n。以後AltキーをCommandキーと読み替えてください)

![class-refs.png](https://qiita-image-store.s3.amazonaws.com/0/61787/f98af469-102f-1f25-3ad7-b325bcb7c752.png)

25箇所ほど`Promise`への参照が見つかります。`PromiseTest`というユニットテストのクラスがあるようなので、それを見るのが使い方を把握するには良さそうです。`PromiseTest`を選択して"browse"をクリックするとブラウザが開きます。

![promise-test-browse.png](https://qiita-image-store.s3.amazonaws.com/0/61787/3ed1df92-163a-9d56-4b3e-43a3ca1acaf4.png)

ユニットテストクラスの場合、ブラウザ上から即座にテストの実行ができます。右クリックメニューから"run all tests"を選ぶと、テストが走り、成功したメソッドがほんのり緑になります。(なんとも地味ですが)

![run-all-tests.png](https://qiita-image-store.s3.amazonaws.com/0/61787/be44f222-a25c-9d51-605a-8fa5b56d23f1.png)

さて、あとはテストメソッド群をつらつらと読んでいけば使い方がわかるでしょう。`Promise`を`new`して何かを実行し、成功なら`resolveWith:`、失敗ならば`rejectWith:`で結果を返すというのが基本のようです。

# どの辺が[Promises/A+](https://promisesaplus.com/)なのか?

つまりはSqueakに元々あった`Promise`に対し、`then:ifRejected:`を追加し、chainableにつなげて書けるようにした、ということのようです。

こんな風に使います。デスクトップメニューで"open"->"workspace"でワークスペースを開き、以下のコードを貼り付けましょう。

```smalltalk

random := 3 atRandom. "1から3までの範囲で乱数生成"
promise := Promise new. "Promise生成"

"3秒後にrandomの値が奇数の時に成功、偶数の時にエラーとなる処理を別スレッドで実行"
[3 seconds wait.
random odd ifTrue: [promise resolveWith: random] ifFalse: [promise rejectWith: (Error new tag: random)] ] fork.	

"結果を待つ。成功の場合はOdd: xxxを表示。失敗の時はエラーダイアログを開く"
promise then: [:ans | Transcript cr; show: 'Odd: ', ans] ifRejected: [:err | err signal: 'even'].

"全体の開始を告げる"
Transcript cr; show: 'Start'
```

実行結果を見るために事前にAlt+tでトランスクリプトを開いておきます。
コードのすべてをAlt+aで選択して、"do it"(Alt+d)しましょう。

まず'Start'と表示されます。別のスレッドで非同期に開始された処理は、3秒後に結果を返してきます。

以下のような感じで表示されるでしょう。

![odd.png](https://qiita-image-store.s3.amazonaws.com/0/61787/3362486c-361e-3449-4eea-5f937a975578.png)

運悪く失敗した場合には、エラーダイアログが立ち上がります。

![even.png](https://qiita-image-store.s3.amazonaws.com/0/61787/f25c03b9-cd5a-45bb-2e8f-42725614dfae.png)

これで基本的な動きは確認できました。

# もう少しJSっぽく拡張してみる

[Promises/A+](https://promisesaplus.com/)の仕様自体は`Promise`の生成方法について何も規定していないので、良いと言えば良いのですが、現状では`new`で生成したpromiseオブジェクトを別スレッドに渡していく必要があり、今ひとつ不便な気がします。

Promisesチェーンを試す前に、[JavaScript版](https://developer.mozilla.org/ja/docs/Web/JavaScript/Reference/Global_Objects/Promise)でのコンストラクタに該当する部分を、拡張メソッドとして追加することにします。

`Promise`をブラウズして、class側を選び、'instance creation'カテゴリを選びます。下のコードペインに次のように書いて、右クリックメニューから"accept"(Alt+s)しましょう。

`Promise class >> executor: aBlock` メソッド

```smalltalk

executor: aBlock
	| promise resolver rejecter |
	promise := self new.
	resolver := [:answer | promise resolveWith: answer].
	rejecter := [:error | promise rejectWith: error].
	[aBlock value: resolver value: rejecter] fork.
	^ promise
```

中身はなんとなくわかりますね。promise を生成しておいて、成功の時はresolverのブロック経由で正常値を、失敗の時はrejecterのブロック経由でエラーを後から渡せるようにしたというだけのことです。これで明示的にpromiseオブジェクトを別スレッド側に渡す必要がなくなります。生成の処理途中でブロッキングされないように`fork`もしています。

![accept.png](https://qiita-image-store.s3.amazonaws.com/0/61787/8f734e20-9485-9283-2a18-81e93245d9f8.png)

なお、"accept"時に開発者のイニシャルを聞かれるかもしれません。適当に答えておきましょう。バージョン管理などに使われるのですが、本記事では触れません。

では、まずは簡単なチェーンを書いてみましょう。ワークスペースに以下をコピペし、"do it"してください。

```smalltalk

promise := Promise executor: [:resolve :reject | 
	| rand |
	2 seconds wait.
	rand := 3 atRandom.
	"奇数で成功、偶数の場合はエラーを返す"
	rand odd ifTrue: [resolve value: rand ] ifFalse: [reject value: (Error new messageText: 'Even')]
].
((promise then: [:ans | ans * 10])
	then:  [:ans | ans + 1])
	then: [:ans | ans inspect ] ifRejected: [:err | err inspect].

Transcript cr; show: 'Promise chaining start'.
```
`then:`が連なっていますね。`rand`が1や3の時は値が引き渡されていき最終結果をinspectできます。

![odd-success.png](https://qiita-image-store.s3.amazonaws.com/0/61787/e5a8d93a-59c0-ddf0-b3b3-479a0aa3a534.png)

2の場合はエラーとなるため、最後の`ifRejected:`部分で拾われて、エラーがinspectされるという結果になります。

![even-error.png](https://qiita-image-store.s3.amazonaws.com/0/61787/b86b3449-d2f6-5638-dcba-fc33020eb6c8.png)

何度か"do it"してもなかなか失敗しないという方は、`ans + 1`の箇所を`ans / 0`にしてみても良いでしょう。たとえ奇数の時も`ZeroDivide`のエラーが途中で起こり、`ifRejected:`で拾われます。

![zero-divide.png](https://qiita-image-store.s3.amazonaws.com/0/61787/588ccd39-4d04-fb03-5c69-4d1f1dc41792.png)

# Promisesチェーンで運試し

`then:`を連ねてみましたが、最初以外はすべて同期処理なので気分が出ません。最後に非同期名処理を連ねて'Happy Squeaking Holidays!!'のメッセージを出したいと思います。ワークスペースに以下を貼り付けて"do it"しましょう。

```smalltalk

delayedMessenger := [:prevMessage :words |
	Promise executor: [:resolve :reject | 
		1 seconds wait.
		nextWord := words atRandom. "候補から適当に選ぶ"
		nextMessage := prevMessage asString, ' ', nextWord. "前の文言とつなげる"
		nextWord = words last 
			ifTrue: [reject value: (Error new messageText: nextMessage)] "選ばれたnextWordが最後の要素だった時には失敗"
			ifFalse: [resolve value: nextMessage]
	].
].

promise := delayedMessenger value: '' value: #('Happy' 'Lucky' 'Wonderful' 'Bad').
((promise then: [:msg | delayedMessenger value: msg value: #('Squeaking' 'Programming' 'Screaming')])
	then: [:msg | delayedMessenger value: msg value: #('Holidays' 'Christmas' 'Monday')])
	then: [:msg | (msg, '!!') inspect] ifRejected: [:error | (error asString, '!?') inspect].

Transcript cr; show: 'Start'
```

'Start'の表示後、3秒ほど時間がかかってインスペクタが開きます。
`executor:`内の処理は非同期で実行されますが、`then:`で同期してメッセージがうまくつながっていくというわけです。

さて、皆さんは何が表示されたでしょうか? よほど運がいいと以下のように表示されます。
![holidays.png](https://qiita-image-store.s3.amazonaws.com/0/61787/fadd5f55-2854-6c32-60d0-515755c2e96e.png)

都合の悪いwordが来ると途中でメッセージがつながらなくなり、エラーとして表示されます。

![sqreaming.png](https://qiita-image-store.s3.amazonaws.com/0/61787/7e7de74c-170f-fcfc-6f39-b84acedbc1d4.png)

# 終わりに

たまにはSqueakも良いものですね。
ここまで来れば[async/await](https://developer.mozilla.org/ja/docs/Web/JavaScript/Reference/Statements/async_function)が使えるように拡張していくのも容易ではと思いますが、それはまた次の機会ということで。